
Introduction

I SAT solving: DPLL with watched literals

I Stability tests in fixpoint calculations

I A solver exploiting delay in Prolog

I Some quick experiments

I Discussion

The DPLL algorithm

(1) function DPLL(f : CNF formula, θ : truth assignment)
(2) begin
(3) θ1 := θ ∪ unit-propagation(f , θ);
(4) if (is-satisfied(f , θ1)) then
(5) return θ1;
(6) else if (is-conflicting(f , θ1)) then
(7) return ⊥;
(8) endif
(9) x := choose-free-variable(f , θ1);
(10) θ2 := DPLL(f , θ1 ∪ {x 7→ true});
(11) if (θ2 6= ⊥) then
(12) return θ2;
(13) else
(14) return DPLL(f , θ1 ∪ {x 7→ false});
(15) endif
(16) end

Unit propagation with Watched Literals

I Where the variables are {u, v ,w , x , y , z}, consider:
¬x ∨ z , u ∨ ¬v ∨ w ,¬w ∨ y ∨ ¬z

I With the partial assignment θ = {x 7→ true} this becomes:
false ∨ z , u ∨ ¬v ∨ w ,¬w ∨ y ∨ ¬z

I For the first clause to be satisfied, the only unassigned
variable z must be assigned to true, and θ is extended with
this, becoming θ′ = {x 7→ true, z 7→ true}
false ∨ true, u ∨ ¬v ∨ w ,¬w ∨ y ∨ false

I It is only necessary to monitor two unassigned variables in a
clause.

I With θ′ extended to θ′′ = {x 7→ true, z 7→ true, y 7→ false},
unit propagation leads to w being assigned to false, but the
second clause does not react to this as w is not monitored
false ∨ true, u ∨ ¬v ∨ false, true ∨ false ∨ false

Background: a Pos-based groundness analyser

I Stability in a fixpoint calculation
might be checked by testing
whether fi |= fi+1 and fi+1 |= fi .

I These entailments become SAT
problems, for example
(x → y) |= x ∨ y becomes cnf
problem ¬x ∨ y ,¬x ,¬y .

I This has satisfying assignment
{x 7→ 0, y 7→ 0}, indicating that
the entailment does not hold.

I Whereas, x |= x ← y becomes
x , y ,¬x . This does not have a
satisfying assignment, hence the
entailment holds.

Delay in Prolog

I Logic Programming = Logic+ Control

I Delay is a fundamental aspect of Control

I It is used to suspend execution until arguments are
appropraitely instantiated:

:- block merge(-,?,-), merge(?,-,-).
merge([], Y, Y).
merge(X, [], X).
merge([H|X], [E|Y], [H|Z]) :- H @< E, merge(X, [E|Y], Z).
merge([H|X], [E|Y], [E|Z]) :- H @>= E, merge([H|X], Y, Z).

I Delays solve the control generation problem: it is always
possible to introduce delays into clauses so as to induce a
terminating control strategy.

I That is, by adding control (delays) to clauses, the logical
specification of an algorithm can be implemented.

Code (SICStus)

sat(Clauses, Vars) :-
problem_setup(Clauses), elim_var(Vars).

elim_var([]).
elim_var([Var | Vars]) :-

elim_var(Vars), (Var = true; Var = false).

problem_setup([]).
problem_setup([Clause | Clauses]) :-

clause_setup(Clause),
problem_setup(Clauses).

clause_setup([Pol-Var | Pairs]) :-
set_watch(Pairs, Var, Pol).

Code (SICStus)

set_watch([], Var, Pol) :- Var = Pol.
set_watch([Pol2-Var2 | Pairs], Var1, Pol1):-

watch(Var1, Pol1, Var2, Pol2, Pairs).

:- block watch(-, ?, -, ?, ?).
watch(Var1, Pol1, Var2, Pol2, Pairs) :-

nonvar(Var1) ->
update_watch(Var1, Pol1, Var2, Pol2, Pairs);
update_watch(Var2, Pol2, Var1, Pol1, Pairs).

update_watch(Var1, Pol1, Var2, Pol2, Pairs) :-
Var1 == Pol1 -> true; set_watch(Pairs, Var2, Pol2).

Example

z x y u v
¬x ∨ ¬y ∨ z
¬x ∨ ¬z ∨ ¬u
¬y ∨ ¬z ∨ u
¬x ∨ ¬v ∨ z
¬x ∨ y ∨ u
y ∨ v ∨ z

false ∨ false ∨ false
Block: sat_engine:watch(_X,false,_Y,false,[true-_Z])

Block: sat_engine:watch(_X,false,_Z,false,[false-_U])

Example

Z 7→ true

Unblock: sat_engine:watch(_X,false,true,false,[false-_U])
Block: sat_engine:watch(_X,false,_U,false,[])

X 7→ true

Unblock: sat_engine:watch(true,false,_Y,false,[true-true])

Unblock: sat_engine:watch(true,false,_U,false,[])

U 7→ false, results in failure
X 7→ false

Unblock: sat_engine:watch(false,false,_Y,false,[true-true])

Unblock: sat_engine:watch(false,false,_U,false,[])

etc...

Delay in other Prolog systems

SWI: when

set_watch([], Var, Pol) :- Var = Pol.
set_watch([Pol2-Var2 | Pairs], Var1, Pol1):-

when(;(nonvar(Var1),nonvar(Var2)),
watch(Var1, Pol1, Var2, Pol2, Pairs)).

watch(Var1, Pol1, Var2, Pol2, Pairs) :- ...

SWI (plus...): freeze

set_watch([], Var, Pol) :- Var = Pol.
set_watch([Pol2-Var2 | Pairs], Var1, Pol1):-

freeze(Var1,V=u), freeze(Var2,V=u),
freeze(V, watch(Var1,Pol1,Var2,Pol2,Pairs)).

watch(Var1, Pol1, Var2, Pol2, Pairs):- ...

Extensions

I Static variable ordering: order variables by frequency of
occurrence in the problem. This wins in two ways: the
problem size is quickly reduced by satisfying clauses and the
amount of propagation achieved is greater.

I Preprocessing with resolution: a popular tactic is to change
the problem by restructuring it using limited applications of
resolution steps.

I Backjumping: allows the solver to avoid exploring fruitless
branches of the search tree.

I Dynamic variables ordering: reorder variables during search.
Reordering can be implemented using similar tactics to
backjumping, but a good implementation also needs learning...

Experiments

benchmark vars clauses sat sics mini assigns

chat 80 1.cnf 13 31 true 0 1 9
chat 80 2.cnf 12 30 true 0 1 5
uf20-0903.cnf 20 91 true 0 1 8
uf50-0429.cnf 50 218 true 10 1 89

uf100-0658.cnf 100 430 true 20 1 176
uf150-046.cnf 150 645 true 290 15 3002
uf250-091.cnf 250 1065 true 2850 171 13920

uuf50-0168.cnf 50 218 false 0 1 79
uuf100-0592.cnf 100 430 false 50 6 535
uuf150-089.cnf 150 645 false 770 18 8394
uuf250-016.cnf 250 1065 false t/o 1970
2bitcomp 5.cnf 125 310 true 130 1 7617
flat200-90.cnf 600 2237 true 380 12 1811

Limitations

I Large problems: the programmer does not have the
fine-grained memory control required to store and access
hundreds of thousands of clauses.

I Learning: clauses are added to the problem that express
regions of the search space that do not contain a solution.
Unfortunately, it is not clear how to achieve this cleanly in this
Prolog solver, as calls to the learnt clauses would be lost on
backtracking.

Conclusions

I A SAT solver can be cleanly and simply implemented in
Prolog using logic: variables and assignment; and control:
unit propagation with watched literal.

I However, the solver will struggle with large problems, owing to
the lack of fine grained memory control required.

I The solver presented provides an easy entry to SAT solving,
and is useful for small to medium sized SAT instances.

